Stylizing your SAS graph – A needs-based approach

Jerome Lechere, Novartis, Basel, Switzerland

ABSTRACT
When producing a SAS® graph, the default rendering is determined by the ODS style associated with the active ODS destination. However SAS allows users to take control over each of the graphical attributes (color, symbol, line pattern…) to tailor results as per their needs. While a quick and partial control might be sufficient to generate a single graph, users may want to take full control over the graph procedures to produce numerous graphs in a consistent manner or to produce complex graphs. This paper should guide readers to use the most appropriate method in order to stylize a SAS graph.

INTRODUCTION
The introduction of the Statistical Graphics (SG) procedures in SAS 9.2 was a great leap forward. Generating graphs became much faster and intuitive. However, the SG procedures came along with many options and statements and their usage might be confusing at first sight. When it comes down to stylizing a graph, several solutions are available to users. We will go through three main methods - looking at their advantages and inconvenience - to help users choose the option that best suits their needs. Before drilling down into the details of each proposed solution, we will have a brief introduction on the SAS ODS styles, which are permanently acting in the background.
BEHIND THE SCENE: SAS ODS STYLES

DEFINITION
An ODS style is a collection of style elements that provides specific graphical attributes for your SAS output. Every ODS output destination has an associated default style. Therefore, your output might look different depending on which destination you use:

- Default style for the PRINTER destination is ‘Printer’
- Default style for the RTF destination is ‘RTF’
- Default style for the HTML and HTML5 destinations is ‘HTMLBlue’

When you issue an ODS statement like:

```
ods rtf body= sample.rtf.
```

You are really issuing the following:

```
ods rtf body= sample.rtf.
style=Default;
```

LIST ODS STYLES
The names of the ODS styles can be accessed using the following program.

```
proc template;
list styles;
run;
```

<table>
<thead>
<tr>
<th>No</th>
<th>Path</th>
<th>Type</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>Styles</td>
<td>Dir</td>
</tr>
<tr>
<td>2</td>
<td>Styles.Analysis</td>
<td>Style</td>
</tr>
<tr>
<td>3</td>
<td>Styles.BlockPrint</td>
<td>Style</td>
</tr>
<tr>
<td>4</td>
<td>Styles.BluePrint</td>
<td>Style</td>
</tr>
<tr>
<td>5</td>
<td>Styles.Default</td>
<td>Style</td>
</tr>
<tr>
<td>6</td>
<td>Styles.Done</td>
<td>Style</td>
</tr>
<tr>
<td>7</td>
<td>Styles.Breeze</td>
<td>Style</td>
</tr>
<tr>
<td>8</td>
<td>Styles.GrayScale</td>
<td>Style</td>
</tr>
<tr>
<td>9</td>
<td>Styles.BluCrest</td>
<td>Style</td>
</tr>
<tr>
<td>10</td>
<td>Styles.FancyPrint</td>
<td>Style</td>
</tr>
<tr>
<td>11</td>
<td>Styles.Festival</td>
<td>Style</td>
</tr>
<tr>
<td>12</td>
<td>Styles.Football</td>
<td>Style</td>
</tr>
<tr>
<td>13</td>
<td>Styles.Gantt</td>
<td>Style</td>
</tr>
<tr>
<td>14</td>
<td>Styles.GrayScalePrinter</td>
<td>Style</td>
</tr>
<tr>
<td>15</td>
<td>Styles.HTMLBlue</td>
<td>Style</td>
</tr>
<tr>
<td>16</td>
<td>Styles.Harvest</td>
<td>Style</td>
</tr>
<tr>
<td>17</td>
<td>Styles.HighContrast</td>
<td>Style</td>
</tr>
<tr>
<td>18</td>
<td>Styles.Journal</td>
<td>Style</td>
</tr>
<tr>
<td>19</td>
<td>Styles.Journal1</td>
<td>Style</td>
</tr>
<tr>
<td>20</td>
<td>Styles.Journal2</td>
<td>Style</td>
</tr>
<tr>
<td>21</td>
<td>Styles.Journal3</td>
<td>Style</td>
</tr>
<tr>
<td>22</td>
<td>Styles.Journal4</td>
<td>Style</td>
</tr>
<tr>
<td>23</td>
<td>Styles.Journal5</td>
<td>Style</td>
</tr>
<tr>
<td>24</td>
<td>Styles.Listing</td>
<td>Style</td>
</tr>
<tr>
<td>25</td>
<td>Styles.Original</td>
<td>Style</td>
</tr>
<tr>
<td>26</td>
<td>Styles.OriginalPrint</td>
<td>Style</td>
</tr>
<tr>
<td>27</td>
<td>Styles.Minimal</td>
<td>Style</td>
</tr>
<tr>
<td>28</td>
<td>Styles.Monochrome</td>
<td>Style</td>
</tr>
<tr>
<td>29</td>
<td>Styles.Monospace</td>
<td>Style</td>
</tr>
<tr>
<td>30</td>
<td>Styles.MoonFlower</td>
<td>Style</td>
</tr>
<tr>
<td>31</td>
<td>Styles.Nebula</td>
<td>Style</td>
</tr>
<tr>
<td>32</td>
<td>Styles.Nostyle</td>
<td>Style</td>
</tr>
<tr>
<td>33</td>
<td>Styles.Normal</td>
<td>Style</td>
</tr>
<tr>
<td>34</td>
<td>Styles.Simple</td>
<td>Style</td>
</tr>
<tr>
<td>35</td>
<td>Styles.Dream</td>
<td>Style</td>
</tr>
<tr>
<td>36</td>
<td>Styles.Ethereal</td>
<td>Style</td>
</tr>
<tr>
<td>37</td>
<td>Styles.Eternity</td>
<td>Style</td>
</tr>
<tr>
<td>38</td>
<td>Styles.PowerPoint</td>
<td>Style</td>
</tr>
<tr>
<td>39</td>
<td>Styles.PowerPointDark</td>
<td>Style</td>
</tr>
<tr>
<td>40</td>
<td>Styles.PowerPointLight</td>
<td>Style</td>
</tr>
<tr>
<td>41</td>
<td>Styles.Printer</td>
<td>Style</td>
</tr>
<tr>
<td>42</td>
<td>Styles.Raven</td>
<td>Style</td>
</tr>
<tr>
<td>43</td>
<td>Styles.Right</td>
<td>Style</td>
</tr>
<tr>
<td>44</td>
<td>Styles.Ruby</td>
<td>Style</td>
</tr>
<tr>
<td>45</td>
<td>Styles.Sapphire</td>
<td>Style</td>
</tr>
<tr>
<td>46</td>
<td>Styles.Seaside</td>
<td>Style</td>
</tr>
<tr>
<td>47</td>
<td>Styles.SeasidePrint</td>
<td>Style</td>
</tr>
<tr>
<td>48</td>
<td>Styles.StarPrint</td>
<td>Style</td>
</tr>
<tr>
<td>49</td>
<td>Styles.Stats</td>
<td>Style</td>
</tr>
<tr>
<td>50</td>
<td>Styles.Starlit</td>
<td>Style</td>
</tr>
<tr>
<td>51</td>
<td>Styles.Starlight</td>
<td>Style</td>
</tr>
<tr>
<td>52</td>
<td>Styles.StarlightDeep</td>
<td>Style</td>
</tr>
</tbody>
</table>
The names and characteristics of the ODS styles can be visualized via ODS Graphics Designer > Tools > Style Editor.

Style = LISTING / Category = LINES / Style element = GRAPHFIT
The GRAPHFIT style element is made of 6 graphical attributes (only 4 visible here)
The FILL COLOR graphical attribute is set to dark blue for the LISTING style

Style = STATISTICAL / Category = LINES / Style element = GRAPHFIT
The GRAPHFIT style element is made of 6 graphical attributes (only 4 visible here)
The FILL COLOR graphical attribute is set to pale blue for the LISTING style
CHANGE THE CURRENT STYLE FOR AN ODS DESTINATION

It requires only the use of the STYLE= option on an ODS destination statement.

ods html style=journal

The journal style is applied to all output for that destination until you change or close the destination or start a new SAS session.

CREATE A NEW STYLE FROM AN EXISTING STYLE

```sas
proc template;
define style styles.TEST; /* New style 'TEST' */
parent = styles.LISTING; /* Existing style 'LISTING' */

style GraphData1 from GraphData1 /
lineStyle = 20
markerSymbol = 'square';

style GraphBackground from GraphBackground /
color = beige;
end;
run;
```

In the parent style 'LISTING',
- GRAPHDATA1 style element,
  - LINESTYLE graphical attribute is set to ‘1’ (solid)
  - MARKERSYMBOL graphical attribute is set to ‘circle’
- GRAPHBACKGROUND style element,
  - COLOR graphical attribute is set to ‘white’

In the newly created style 'TEST',
- GRAPHDATA1 style element,
  - LINESTYLE graphical attribute is set to ‘20’ (dash)
  - MARKERSYMBOL graphical attribute is set to ‘square’
- GRAPHBACKGROUND style element,
  - COLOR graphical attribute is set to ‘beige’
CHANCE A STYLE ELEMENT IN THE CURRENT STYLE

In the below example below, the default ‘Graphfit’ style element is changed by the ‘Graphfit2’ style element. As a consequence, the default graphical attributes:
- line color is changed from dark blue to red
- line pattern is changed from solid to shortdash

- Specify a different style element with ODS Graphics Designer:

- Specify a different style element from within a SG procedure:
  
  ```
  proc sgplot data=sashelp.class;
  density height / lineattrs=graphfit2;
  run;
  ```

- Specify a different style element from within a GTL proc template:
  
  ```
  proc template;
  define statgraph style_element;
  begingraph /;
  <...>
  scatterplot x=XVAR y=YVAR / lineattrs=graphfit2;
  <...>
  endgraph;
  end;
  run;
  ```
- Change style element from ‘Graphfit’ to ‘Graphfit2’

**Style element** Graphfit / line color = dark blue / line pattern = solid

![Graphfit: dark blue line pattern = solid](image)

**Style element** Graphfit2 / line color = red / line pattern = shortdash

![Graphfit2: red line pattern = shortdash](image)
CHANGE A GRAPHICAL ATTRIBUTE IN A CURRENT STYLE ELEMENT

In the below example below, the default graphical attributes ‘color’ is changed from dark blue to red.

- Specify a different graphical attribute with ODS Graphics Designer:

- Specify a different graphical attribute from within a SG procedure:

```sas
proc sgplot data=sashelp.class;
density height / lineattrs=(color=red);
run;
```

- Specify a different graphical attribute from within a GTL proc template:

```sas
proc template;
define statgraph app_attr;
begingraph /;
<...>
DensityPlot 'Height'n / lineattrs=(color=red);
<...>
endgraph;
end;
run;
```
• Change graphical attribute 'color' in the current Graphfit style element

Style element Graphfit / line color = dark blue

Style element Graphfit / line color = red

All the other graphical attributes from the Graphfit style element remain the same (solid line pattern for instance).
STYLIZING A SAS GRAPH USING THE ATTRIBUTE OPTIONS

DEFINITION
An attribute option allows the user to control a graph attribute from within a SG procedure or a GTL proc template. The option is introduced by a slash '/' and its attributes must be enclosed in parentheses and specified as a name=value pair.

```
needleplot x=x y=y / Lineattrs = (pattern=1 thickness=8);
scatterplot x=x y=y / Markerattrs = (symbol=StarFilled);
barchart category=x response=y / Fillattrs = (transparency=0.5);
```

SYNTAX IN A SG PROCEDURE
```
proc sgplot data=input;
   needle x=x y=y / Lineattrs=(pattern=1 thickness=8);
run;
```

SYNTAX IN A GTL PROC TEMPLATE
```
proc template;
define statgraph wfp;
begingraph /;
   <...
   needleplot x=x y=y / Lineattrs=(pattern=1 thickness=8);
   <...
endgraph;
end;
run;
```

```
proc sgrender
data=input
template=wfp;
run;
```
ILLUSTRATED EXAMPLE

- Original plot with default attributes: color = blue / thickness = 1

```sas
proc template;
define statgraph wfp;
begingraph /;
EntryTitle "Figure 1a - Best percentage change from baseline in criteria, by treatment" /;
layout overlay /
  xaxisopts=(Label="Subject number" LABELATTRS=(size=10) display=(label ticks tickvalues) TICKVALUEATTRS=(size=8) type=linear linearopts=(tck_idx= tickvaluelist=(1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17 18 19 20 21 22 23 24 25 26 27 28 29 30 31 32 33 34 35 36 37 38 39 40 41 42 43) viewmin= 0.5 viewmax=43.5 TICKVALUEFITPOLICY=ROTATETHIN)
  yaxisopts=( Label="Best % change from baseline" LABELATTRS=(size=10) type=linear TICKVALUEATTRS=(size=7) linearopts=( tickvaluelist=( -100 -80 -60 -40 -20 0 20 40 60 80 100 ) viewmin=-120 viewmax=120 )
    needleplot X=x Y=y /Lineattrs=(pattern=1) name='wfp';
Entry halign=right TEXT /valign=top textattrs=(size=9);
ReferenceLine y=0 / clip=true Lineattrs=( Color=CX000000);
ReferenceLine y=-30 / clip=true Lineattrs=( Color=CX000000 PATTERN=4);
endlayout;
endgraph;
end;
run;
proc sgrender
data=input
template=wfp;
run;
```

Figure 1a - Best percentage change from baseline in criteria, by treatment
DEFAULT ATTRIBUTES: COLOR = BLUE - THICKNESS = 1
• Updated plot with new attributes: color = green / thickness = 16

```
proc template;
define statgraph wfp;
begingraph /;
   <...>
   needleplot X=x Y=y / Lineattrs=(pattern=1 color=green thickness=16);
   <...>
endgraph;
end;
run;
```

The newly defined graphical attributes ‘Color = green’ and ‘Thickness = 16’ apply for ALL values in the plot.
IN SUMMARY:

- WITH THE ATTRIBUTE OPTIONS, THE GRAPHICAL ATTRIBUTE DEFINED BY USERS IS APPLIED TO ALL VALUES IN THE PLOT. THEREFORE, USING ATTRIBUTE OPTIONS IS ACCEPTABLE IF NO GROUP VARIABLE IS INVOLVED BUT IS NOT RECOMMENDED OTHERWISE.

- ACTUALLY, IF A GROUP VARIABLE IS INVOLVED, SAS TRIES ITS BEST TO DIFFERENTIATE THE DIFFERENT GROUPS BY CYCLING THROUGH THE GRAPHICAL ATTRIBUTES FROM THE ODS STYLE:
  - if users forces ‘symbol=StarFilled’ SAS will apply different colors to differentiate the groups
  - if users forces ‘color=Blue’ SAS will apply different symbols to differentiate the groups
  - if users forces both ‘symbol=StarFilled’ and ‘color=Blue’, SAS can no longer differentiate the groups. All values on the plot will be represented by an identical Blue + StarFilled marker

- AS A CONSEQUENCE, IF A GROUP VARIABLE IS INVOLVED AND USERS WANT TO TAKE CONTROL OVER THE GRAPHICAL ATTRIBUTES, IT IS STRONGLY RECOMMENDED TO USE THE STYLEATTRS STATEMENT.
STYLIZING A SAS GRAPH USING THE STYLEATTRS STATEMENT

DEFINITION
Styleattrs statement allows temporary editing of the default ODS style.
Datacolors / datacontrastcolors / datalinepatterns / datasymbols are options for the styleattrs statement.
When producing a graph where a group variable is involved:

- By default, SAS access the ODS style associated to the open ODS destination and read its attributes in sequential order. For instance, if ODS destination is HTML, SAS access the 'HTMLBlue' style and render the group 1 values in dark blue, the group 2 values in dark red and the group 3 values in dark green.
- With the Styleattrs statement, users are able to hardcode the sequence of the attributes directly in the code, for instance group 1 values in dark green, group 2 values in dark red and group 3 values in dark blue.

The group variable is define using the Group= options.

Warning 1: ODS graphics must be on for SAS to know the style attached to the open ODS destination and then modify it accordingly.
Warning 2: For SAS to execute the sequence of attributes hardcoded in the Styleattrs statement, the attrpriority option must be turned off (attrpriority=none).

SYNTAX IN A SG PROCEDURE

```
ods graphics / attrpriority=none;
proc sgplot data=input;
  styleattrs dataContrastColors = ( CXADD8E6  CX98FB98  CXF08080 ) /* LightBlue PaleGreen LightCoral */
    dataSymbols = ( CIRCLE SQUARE );
  needle x=x y=y / group=trt_grp;
run;
```

SYNTAX IN A GTL PROC TEMPLATE

```
proc template;
  define statgraph wfp;
  begingraph / dataContrastColors=(  CXADD8E6  CX98FB98  CXF08080 ) /* LightBlue PaleGreen LightCoral */
    dataSymbols=( CIRCLE SQUARE );
    <...>
    needleplot x=x y=y / group=trt_grp;
    <...>
  endgraph;
end;
run;
```

```
ods graphics / attrpriority=none;
proc sgrender
data=input
template=wfp;
    run;
```
ILLUSTRATED EXAMPLE

```sas
proc template;
define statgraph wfp;
begingraph / dataContrastColors=( CXADD8E6 CX98FB98 CXF08080 ) /* LightBlue PaleGreen LightCoral */
   dataSymbols=( CIRCLE SQUARE );
<...>
needleplot x=x y=y / group=trt grp;
<...>
endgraph;
end;
run;

ods graphics / attrpriority=none;
proc sgrender
   data=input
   template=wfp;
run;
```

- Original plot. The input dataset shows subjects enrolled into the:
  - INV. DRUG 10MG group
  - INV. DRUG 20MG group

![Figure 2a - Best percentage change from baseline in criteria, by treatment](image-url)
Updated plot. The input dataset now shows subjects enrolled into the:
- INV. DRUG 10MG group
- INV. DRUG 15MG group
- INV. DRUG 20MG group

The color assignment is based on the sequence of color hardcoded into the `styleattrs` statement: first treatment group in the dataset represented in light blue, second treatment group in the dataset represented in pale green, third treatment group in the dataset represented in light coral and so on.

In the first plot, subjects enrolled into the INV. DRUG 20MG group (second treatment group in the dataset) are represented in pale green (second color listed in the `styleattrs` statement).

In the first plot, subjects enrolled into the INV. DRUG 20MG group (third treatment group in the dataset) are represented in light coral (third color listed in the `styleattrs` statement).

The plot rendering is dependent from the group variable in the input dataset (sorting order, missing/existing values).
IN SUMMARY

❖ WHENEVER A GROUP VARIABLE IS INVOLVED AND USERS WANT TO DIFFERENTIATE THE DIFFERENT ‘BY’ GROUPS IN THE PLOT, THE ATTRIBUTE OPTIONS MUST NOT BE USED AND THE STYLEATTRS STATEMENT SHOULD BE CONSIDERED.

❖ WITH THE STYLEATTRS STATEMENT, USERS ARE ABLE TO HARDCODE A DESIRED SEQUENCE OF GRAPHICAL ATTRIBUTES DIRECTLY INTO A SG PROCEDURE OR A GTL PROC TEMPLATE.

❖ THESE ATTRIBUTES ARE APPLIED TO THE INPUT DATA AS PER THE FOLLOWING RULE:
  • First graphical attribute ☸ Group variable, first value in the dataset
  • Second graphical attribute ☸ Group variable, second value in the dataset
  • Third graphical attribute ☸ Group variable, third value in the dataset

❖ USING THE STYLEATTRS STATEMENT ALLOWS USERS TO STYLIZE MANY ASPECTS OF A SAS GRAPH IN AN INTUITIVE AND QUICK MANNER. HOWEVER THE RESULT IS DEPENDENT FROM THE INPUT DATASET.

❖ FOR USERS TO TAKE FULL CONTROL OVER THEIR SAS GRAPH, IT IS STRONGLY RECOMMENDED TO CREATE AND USE AN ATTRIBUTE MAP DATASET.
STYLIZING A SAS GRAPH USING AN ATTRIBUTE MAP DATASET

DEFINITION
An attribute map dataset is a specific dataset holding all the required information to associate graphical attributes to values from the input dataset. This dataset must include the following 2 reserved variables:

- ID = attribute map identifier. ID identifies each attribute map. An attribute map dataset can contain more than one attribute map. This capability enables users to apply different attribute maps to different group variables in a plot. The example below shows 2 attribute map: ‘trt’ for treatment and ‘resp’ for response.
- VALUE = group value. This variable lists the different possible values for a define group variable. The example below shows that the treatment group variable can take 3 different values in the input dataset: ‘INV. DRUG 10MG’, ‘INV. DRUG 15MG’ and ‘INV. DRUG 20MG’. The response group variable can take 5 different values in the input dataset: ‘CR’, ‘PR’, ‘SD’, ‘PD’ and ‘UNK’.

Beside ID and VALUE, users need to define the attribute variables of interest such as FILLCOLOR=, LINECOLOR=, LINEPATTERN=, MARKERSYMBOL, and so on. The example below shows that the different values for the treatment variable will be differentiated in the plot by using different line colors.

SYNTAX TO CREATE AN ATTRIBUTE MAP DATASET
The most commonly used method for creating an attribute map dataset is via a data step using the ‘datalines’ statement:

```sas
data attrds;
  infile datalines delimiter=';' missover;
  length ID $8 VALUE $40 linecolor $12 markersymbol $40 ;
  input ID VALUE linecolor markersymbol;
  datalines;
  trt,INV. DRUG 10MG,LightBlue
  trt,INV. DRUG 15MG,LightCoral
  trt,INV. DRUG 20MG,PaleGreen
  resp,CR, ,starfilled
  resp,PR, ,star
  resp,SD, ,circle
  resp,PD, ,trianglefilled
  resp,UNK, ,hash
; run;
```

<table>
<thead>
<tr>
<th>ID</th>
<th>VALUE</th>
<th>linecolor</th>
<th>markersymbol</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>trt</td>
<td>INV. DRUG 10MG</td>
<td>LightBlue</td>
</tr>
<tr>
<td>2</td>
<td>trt</td>
<td>INV. DRUG 15MG</td>
<td>LightCoral</td>
</tr>
<tr>
<td>3</td>
<td>trt</td>
<td>INV. DRUG 20MG</td>
<td>PaleGreen</td>
</tr>
<tr>
<td>4</td>
<td>resp</td>
<td>CR</td>
<td></td>
</tr>
<tr>
<td>5</td>
<td>resp</td>
<td>PR</td>
<td></td>
</tr>
<tr>
<td>6</td>
<td>resp</td>
<td>SD</td>
<td></td>
</tr>
<tr>
<td>7</td>
<td>resp</td>
<td>PD</td>
<td></td>
</tr>
<tr>
<td>8</td>
<td>resp</td>
<td>UNK</td>
<td></td>
</tr>
</tbody>
</table>

Each observation defines the attribute(s) for a particular value of a group variable e.g. Observation 1 indicates that each value from the input dataset where treatment = ‘INV. DRUG 10MG’ will be rendered in a light blue color on the plot.
SYNTAX TO ACCESS AN ATTRIBUTE MAP DATASET FROM WITHIN A SG PROCEDURE

```
proc sgplot data=input datattrmap=attrds;
  needle x=x y=y / group=trt_grp attrid=trt;
run;
```

SYNTAX TO ACCESS AN ATTRIBUTE MAP DATASET FROM WITHIN A GTL PROC TEMPLATE

```
proc template;
  define statgraph wfp;
  begingraph /;
    <...
    needleplot x=x y=y / group=trt_grp;
    <...
  endgraph;
  end;
run;
```

```
proc sgrender
data=input
template=wfp
dattrmap=attrds;
dattrvar trt_grp="trt";
run;
```
ILLUSTRATED EXAMPLE

```plaintext
proc template;
define statgraph wfp;
begingraph /
<...>
needleplot x=x y=y / group=trt_grp ;
<...>
endgraph;
end;
run;

proc sgrender
  data=input
template=wfp
dattrmap=attrds;
dattrvar trt_grp="trt";
run;
```

- Original plot. The input dataset shows subjects enrolled into the:
  - INV. DRUG 10MG group
  - INV. DRUG 20MG group

Figure 3a - Best percentage change from baseline in criteria, by treatment
ATTRIBUTE MAP DATASET - 2 TREATMENT GROUPS

Best % change from baseline

15 9 19 10 18 30 21 32 12 38 28 31 27 11 7 29 4 17
Subject number

INV. DRUG 10MG  INV. DRUG 20MG
• Updated plot. The input dataset now shows subjects enrolled into the:
  - INV. DRUG 10MG group
  - INV. DRUG 15MG group
  - INV. DRUG 20MG group

Figure 3b - Best percentage change from baseline in criteria, by treatment
ATTRIBUTE MAP DATASET - 3 TREATMENT GROUPS

- The color assignment is based on the attribute map dataset: INV. DRUG 10MG is assigned to color light blue, INV. DRUG 15MG is assigned to color light coral and INV. DRUG 20MG is assigned to color pale green.
- Whether the input dataset has 2 values for the 'by' variable (subjects enrolled into the INV. DRUG 10MG group + INV. DRUG 20MG group) or 3 values for the 'by' variable (subjects enrolled into the INV. DRUG 10MG group + INV. DRUG 15MG group + INV. DRUG 20MG group) does not impact the color assignment in the plot.
- The plot rendering is independent from the group variable in the input dataset.
IN SUMMARY

✶ WHENEVER A GROUP VARIABLE IS INVOLVED AND USERS WANT TO CONTROL DISPLAY OF THE DIFFERENT ‘BY’ GROUPS IN THE PLOT, THE CREATION OF AN ATTRIBUTE MAP DATASET SHOULD BE CONSIDERED.

✶ AN ATTRIBUTE MAP DATASET ENABLES USERS TO HARDCODE A LIST OF GRAPHICAL ATTRIBUTES, HARDCODE A LIST OF POSSIBLE VALUES FOR THE ‘BY’ VARIABLE AND PAIR THEM ONE BY ONE IN A FIXED MANNER:

- Graphical attribute X ⇔ Group variable, value A
- Graphical attribute Y ⇔ Group variable, value B
- Graphical attribute Z ⇔ Group variable, value C

✶ THE GRAPHICAL ATTRIBUTES ASSIGNMENT WILL ALWAYS BE THE SAME, IRRESPECTIVE OF THE INPUT DATASET.

✶ THEREFORE, WHILE ATTRIBUTE OPTIONS AND STYLEATTRS STATEMENT ARE TECHNICS TO BE USED ON A PLOT BY PLOT BASIS ONLY, THE ATTRIBUTE MAP DATASET IS THE RECOMMENDED APPROACH WHEN USERS NEED TO PRODUCE SEVERAL GRAPHICS ON A PROJECT/STUDY AND WANT TO ENSURE THAT THE SAME ATTRIBUTES WILL BE USED CONSISTENTLY ACROSS ALL PLOTS.
CONCLUSION
Looking at a few SAS versions back, it used to be quite tedious to control graphical attributes, creating for instance an ANNOTATE dataset, or most recently using the TEMPLATE procedure to create new tailored ODS styles. With SAS 9.2, it became easier to stylize a graph. Several options are available to users:

- A quick and easy manner to proceed is to use an attribute option (e.g. lineattrs) from within the SG procedures.
- For graphs where a group variable is involved and users need to distinguish each group in the plot, it is recommended to abandon the attribute options to the favor of the Styleattrs statement.
- Finally, users are even given the possibility to externalize the graphical attributes information into a dedicated dataset (Attribute Map Dataset) in order to take full control over the graph appearance.

REFERENCES
2. http://support.sas.com/documentation/cdl/en/odsug/67921/HTML/default/viewer.htm#n1ooja9ez1lrrn1s2bilfpywp03.htm

ACKNOWLEDGMENTS
Thank you to Nassim Sleiman and Mathias Lebreton for their contributions to this paper.

CONTACT INFORMATION
Your comments and questions are valued and encouraged. Contact the author at:

Jerome Lechere
Senior Statistical Programmer
Oncology Biometrics and Data Management
Novartis Pharma AG
Postfach CH-4002 Basel
Switzerland
Work Phone: +41 61 69 61232
Email: jerome.lechere@novartis.com
Web: www.novartis.com

Brand and product names are trademarks of their respective companies.